


再帰型プログラミングの試み

1. Example of Peano curve

区間 $[0, 1]$ において定義される連続関数 $f(t)$ および $g(t)$ によって, xy 平面に座標 $(f(t), g(t))$ をとると t が 0 から 1 まで連続に変化すると座標点 $(f(0), g(0))$ から $(f(1), g(1))$ まで $(f(0), g(0))$ 連結する右図のような曲線が得られる.



通常, 平面上の 1 点が $(f(t), g(t))$ に対応するけれども, 以下に示すように, 平面のある区間が $(f(t), g(t))$ に対応するとして描かれるのがペアノ曲線¹である. 例えば, 図 1 左のように三角形を分割し 4 個の領域に分ける. 区分された三角形を更に 4 個に分割を繰り返す. 得られた各小三角形の重心は図 1 中に示されている. これらの点 16 個を次々につなぐと図 1 右のような曲線が得られる. これがペアノ曲線の一例である.




図 1: ペアノ曲線 of 例

以下のような Full BASIC²による再帰型 SUB プログラム Peano(n)³によって容易にペアノ曲線が得られる. 図 2 は, 正方形の対角線で得られる 2 つの三角形のそれぞれを 5 回分割を繰り返して得られたものである. この閉じた曲線をシェルピンスキー曲線という. 閉じた曲線を示すために線の内部が塗りつぶされている.

```
OPTION ANGLE DEGREES
PICTURE Peano(n)
  IF n=1 THEN
    PLOT LINES: 1.0, 0.3333 ;
  ELSE
    DRAW Peano(n-1) WITH SCALE(1/2)
    DRAW Peano(n-1) WITH SCALE(1/2)*ROTATE(90)*SHIFT(1,0)
    DRAW Peano(n-1) WITH SCALE(1/2)*ROTATE(-90)*SHIFT(1,1)
    DRAW Peano(n-1) WITH SCALE(1/2)*SHIFT(1,0)
  END IF
END PICTURE
SET WINDOW -0.1,1.6,-0.1,1.6
SET BEAM MODE "IMMORTAL"
LET s2=SQR(2)
DRAW Peano(5) WITH ROTATE(45)
DRAW Peano(5) WITH ROTATE(-135)*SHIFT(s2,s2)
PLOT LINES : 0.0295,0.059
PAINT 0.032,0.059
END
```




図 2: シェルピンスキー曲線より得られる模様

¹Giuseppe Peano (1858 ~ 1932), イタリアの数学者, 記号論理学の開拓者. 1890 年に彼によって提案された.

²参照: <http://hp.vector.co.jp/authors/VA008683/>

³上の Home Page にあるインストラ版を展開すると, 各種プログラムが「SAMPLE」等から得ることができる.

2. Hilbert curve

ヒルベルト曲線は次のような考察によって、ペアノ曲線と同様に再帰型 SUB プログラムを実行することによって得られる。図 3 のように正方形の領域を 4 分割し、各小正方形の中心を印 \circ から始め印 \times まで線を引きコの字型を描く。中心には対称性を明示するためにアルファベットの「R」が示されている。分割を 1 段階進めると図 4 が得られる。各小正方形の中心には軌跡と同じ対称関係にある「R」が示されている。




図 3: 第 1 段階の分割での軌跡




図 4: 第 2 段階の分割と第 1 段階との対称関係




図 5: 6 段階進めたヒルベルト曲線

図 5 は 6 段階、分割を進めて描かれたものである。それを描くための Full BASIC プログラムは以下のごとくである。再帰型 SUB プログラムには上に示された「R」の対称関係が考慮されている。

```
OPTION ANGLE DEGREES
PICTURE Hilbert(n)
  IF n=1 THEN
    PLOT LINES: 0,0 ;
  ELSE
    DRAW Hilbert(n-1) WITH SCALE(1/2,-1/2)*ROTATE(90)*SHIFT(-1,-1)
    DRAW Hilbert(n-1) WITH SCALE(1/2)*SHIFT(-1,1)
    DRAW Hilbert(n-1) WITH SCALE(1/2)*SHIFT(1,1)
    DRAW Hilbert(n-1) WITH SCALE(1/2,-1/2)*ROTATE(-90)*SHIFT(1,-1)
  END IF
END PICTURE
SET WINDOW -2.1,2.1,-2.1,2.1
SET BEAM MODE "IMMORTAL"
DRAW Hilbert(6)
END
```